



LETTERS op het grafisch tv-display



DEEL 2

D. M. DE BOER

In het vorige deel zijn we uitgebreid ingegaan op de manier waarop de karakters in het geheugen staan, de instelmogelijkheden van het programma, en de wijze waarop het programma karakters spiegelt, draait en/of schuift. In dit deel zullen we het stroomdiagram en het programma voor de 6502 behandelen.

Het initialiseren (afb. 2)

Bij het initialiseren krijgen de diverse hulpvariabelen hun beginwaarde. Omdat het programma is uitgevoerd als subroutine gaan we ervan uit dat de richtingsregisters van de PIA's reeds goed staan. In afb. 1 ziet u nog eens welke aansluitingen ingang zijn en welke als uitgang dienst doen. Het initialiseren begint met het destilleren van de gewenste voor- en achtergrondkleur uit geheugenplaats 'KLEUR'. De voorgrondkleur komt in 'VOOR' te staan, de achtergrond in 'ACHTER'. Als bit 0 van 'KLEUR' de waarde '0' heeft zal het programma de lengte, de stuurbits en het beginadres van het karakter bepalen aan de hand van de gegevens uit de tabel. De lengte en de stuurbits



afb. 1 De aansluitingen op de PIA.

afb. 2 Het eerste stukje van het stroomdiagram. Hier krijgen de diverse geheugens hun beginwaarde.


worden rechtstreeks uit de tabel gehaald en op resp. 'LENGTE' en 'MODE' gezet. Het beginadres van het karakter wordt bepaald door de lengten van alle voorgaande karakters op te tellen bij het beginadres van de karakterset. We zijn inmiddels aangeland op adres 0646 (afb. 2 lijst 1), en op dit moment kennen we alle gegevens van het karakter. We beginnen nu met het berekenen en instellen van de hulpvari-

abelen. Allereerst YGRL en YGRH. Met deze coördinaten is een balk gedefinieerd van 10 puntjes hoog, waartussen alle (normale) karakters komen te liggen. Deze coördinaten worden later in het programma gebruikt om de achtergrond boven, onder en tussen de letters aan te vullen.

Op adres 065E wordt bepaald of het karakter al dan niet geschoven is. Wanneer het karakter geschoven is wordt YCOR overeenkomstig ingesteld. Later in het programma wordt hij ook weer terug gezet, zodat YCOR uiteindelijk ongewijzigd uit de strijd komt. Om snelheidsredenen wordt straks bij het tekenen van één byte steeds alléén de Y-coördinaat in het display veranderd bij normale karakters, en steeds alléén de X-coördinaat bij gekantelde karakters. Daarom worden nu reeds zowel het X adres als het Y adres in het display gezet (adres 0666). De laatste stap is het instellen van de hulpcoördinaten YCOR2, YCOR3, XCOR2, BREED2 en HOOG2. Zoals in het vorige deel al ter sprake kwam worden deze hulpcoördinaten bij gekantelde karakters 'gedraaid' overgenomen. De coördinaten XCOR2 en YCOR2 zijn de 'lopende' coördinaten. YCOR3 blijft tijdens het tekenen ongewijzigd, en wordt gebruikt om na het tekenen van 8 bitjes steeds YCOR2 weer op de oude waarde te kunnen zetten.

Het tekenen van een karakter

In afb. 3 en 4 ziet u het stroomdiagram van de kern van het programma. Afb. 3 geeft de procedure voor het tekenen van één byte uit de karakterset, afb. 4 zorgt dat steeds een nieuw karakterbyte wordt aangevoerd. Omdat elk karakter zowel in de breedte als in de hoogte kan worden uitgerekt zijn er wat lus-



sen en tellers nodig. Elk byte wordt vertikaal neergezet (uitgaande van de normale situatie), met MSB onder. Wanneer we een karakter 3x zo breed willen hebben, moet elk byte 3x worden getekend. Daarom moeten we aan het begin van dit programma een teller instellen die bijhoudt hoeveel maal hetzelfde byte is getekend. Deze teller heeft de naam 'BREED3' gekregen. In het volgende blokje (06AC) wordt geheugenplaats 'TEL' op 08 gezet. Met deze teller houden we bij of er al 8 bitjes van het byte zijn getekend. Indexregister Y wordt gebruikt om het te tekenen byte aan te wijzen en staat nu op 0. Op adres 06BO wordt hierdoor het eerst te tekenen byte opgehaald en op 'TEMP' gezet. De laatste in te stellen teller is de teller voor de verhogingsfactor. Hiervoor gebruiken we indexregister X. Na alle instellingen gaan we het eerste bitje uit het te tekenen byte schuiven. Normaal schuiven we naar links (MSB eerst). Alleen als we met het spiegelbeeld van een gekanteld figuur bezig zijn, moeten we naar rechts schuiven. Bit 6 van 'VLAG' geeft deze situatie aan. In ieder geval staat na het schuiven het te tekenen bitje in de carry.

Tekenen een punt in de gewenste kleur

We komen nu in het hart van alle lussen (afb. 3). Omdat dit programmadeel veel herhaald wordt is dit stukje programma ontworpen op snelheid en niet zozeer op een klein aantal programmastappen. Bij het tekenen van één stip kunnen zich de volgende situaties voordoen: De stip moet worden getekend in de voorgrondkleur (carry=1). De stip moet worden getekend in de achtergrondkleur (carry=0), de stip moet niet getekend worden (bit 6 of 7 van 'KLEUR' = 0 en carry = 0 resp. 1). In de eerste twee gevallen wordt de kleur ingesteld, waarna naar adres 06CB wordt gesprongen. Hier wordt alléén het nieuwe Y-adres ingeklokt, omdat het X-adres gedurende het tekenen van een byte ongewijzigd blijft. (Voor een gekanteld karakter wordt uiteraard alléén het X-adres ingeklokt) Vervolgens wordt de schrijfpuls gegeven en de stip is getekend. Wanneer de stip niet getekend moest worden, omdat bit 6 of 7 (6 voor de achtergrond en 7 voor de voorgrond) '0' was, wordt YCOR2 zó ingesteld dat hij de plaats aanwijst van het eerstvolgende te

afb. 3 Dit stuk stroomdiagram tekent één byte in de gewenste hoogte- en breedte.


afb. 4 Het gedeelte dat zorg draagt voor het eventueel spiegelen van het karakter.

tekenen bitje. In alle gevallen geldt dat de verhogingsfactor (staat op dit moment in indexregister X) aangeeft hoeveel maal een stip herhaald moet worden, vóórdat het volgende bit uit het te tekenen byte geschoven wordt. Dit wordt aangegeven door de binnenste lus (afb. 3). Steeds als deze lus wordt doorlopen, wordt register X met één verminderd. Pas als dit register 0 is, wordt op adres 06F0 gekeken of er nog een bit van dit byte getekend moet worden. Zolang 'TEL' (welke we hadden ingesteld op 8) nog geen '0' is, wordt het te tekenen byte nog een keer geschoven.

Natuurlijk wordt ook de verhogingsfactor opnieuw ingesteld, omdat ook het nieuwe bitje misschien een aantal keren boven elkaar zal moeten worden getekend. Wanneer het byte is 'leeg geschoven' moeten we de coördinaten weer goed zetten voor de volgende kolom. Dit gebeurt vanaf adres 06F4. YCOR2 krijgt hier weer zijn oorspronkelijke waarde, terwijl XCOR2 met één wordt verhoogd. Vervolgens worden X- en Y-adres in het display gezet, zonder een schrijfpuls te genereren. Bij karakters in de normale positie behoeft alleen het X-adres ingeklokt te worden (XCOR2 X-adres), omdat het Y-adres voor elk bit opnieuw wordt ingeklokt. Bij een gedraaid karakter is dit net andersom, hier moet XCOR2 in het Y-buffer geheugen worden gezet, terwijl het X-adres voor elk bit opnieuw wordt ingeklokt. We belanden nu bij de laatste lus in afb. 3. Hetzelfde byte moet net zoveel keer getekend worden als de verbredingsfactor aangeeft. Op adres 0714 wordt gekeken of de teller 'BREED3' al bij 0 is. Zolang dit niet het geval is wordt de hele voorgaande procedure herhaald, zij het één positie verder op het beeld. Zodra er een nieuw byte nodig is belanden we in afb. 4.

Het bepalen van het volgende byte

In afb. 4 ziet u het stukje flowdiagram dat niet alleen bepaalt wélk byte nu getekend moet worden, maar bovendien alle handelingen verricht om een



4 karakter te spiegelen (symmetrie). We maken onderscheid tussen 'rechte' symmetrie (zoals bij de A en de I, zie het vorige deel) en 'gekantelde' symmetrie (zoals bij de schaaktoren). De rechte symmetrie wordt aangegeven door bit 7 van 'VLAG', de gekantelde symmetrie door bit 6. We komen in afb. 4 op adres 0740 binnen. Zolang nog niet alle bytes uit de karakterset getekend zijn, wordt op adres 0744 het

volgende byte uit de karakterset ingesteld. Hierna springen we terug naar afb. 3, waar het nieuwe byte in de juiste afmetingen getekend wordt. Zodra het laatste byte getekend is komen we via 0749 op adres 074D, waar gekeken wordt of het karakter gespiegeld moet worden. Wanneer dit niet het geval is rollen we door naar afb. 5. Als symmetrie wel wordt gevraagd zetten we voorlopig bit 7 van 'VLAG' ten teken

van rechte symmetrie. Op 0756 wordt nu gekeken of het figuur ook werkelijk rechtop staat. We nemen voorlopig aan dat dit zo is. De laatste voorwaardelijke sprong vinden we op 075E. Hier wordt gekeken of de middenpoot dubbel moet worden genomen zoals bijvoorbeeld bij de A of enkel zoals bijvoorbeeld bij de I (zie het vorige deel). Bij een dubbele middenpoot springen we weer terug naar afb. 3 waar het hele proces op exact dezelfde wijze wordt herhaald. Het enige verschil met de eerste keer is dat nu bit 7 van 'VLAG' log 1 is. Hierdoor is op adres 0740 (bovenaan in afb. 4) de 'wissel' omgezet, en wordt steeds het vorige karakterbyte ingesteld. Dit proces gaat door tot het 1e byte van het karakter uit de karakterset is getekend. Op dit moment is het gespiegelde gedeelte ook klaar en gaan we naar afb. 5. Bij het laatste geval, de gekantelde symmetrie gaan we bij 0756 (afb. 4) naar 0765. Hier worden de hulpcoördinaten ingesteld op de eerstvolgende karakterplaats, zij het zonder tussenruimte. Geheugenplaats 'VLAG' wordt nu ingesteld op gekantelde symmetrie, en register Y wordt 0 gemaakt, zodat hij weer het eerste karakterbyte aanwijst. Doordat 'VLAG' ingesteld is op gekantelde symmetrie wordt de wissel op adres 0740 nu niet bediend (het programma neemt dus steeds gewoon het volgende byte). Wel wordt steeds het karakterbyte naar rechts geschoven in plaats van naar links, waardoor bit 0 het eerst getekend wordt. Als op deze manier het spiegelbeeld tot stand is gekomen gaan we via 0745 en 0749 naar afb. 5.

Het opvullen van de ruimte tussen de letters

Het belangrijkste werk is gedaan, de letter is getekend. We zijn er echter nog niet, door mogelijkheden van kanten en schuiven hebben niet alle letters dezelfde hoogte. Wanneer we de achtergrond willen tekenen moeten alle beeldstippen tussen YGRL en YGRH, waar geen letter zit, de achtergrondkleur krijgen. Alleen dan worden alle resten van oude tekst en/of tekeningen overschreven (zie afb. 3 vorig deel). Het opvullen van de achtergrond gebeurt volgens afb. 5. Alléén gekantelde karakters kunnen hoger dan 8 punten zijn. In dit geval kan de ruimte boven de letter niet worden opgevuld (07C2).



SOFTWARE grafisch display

In afb. 6 ziet u het laatste stuk van het flowdiagram. Hier worden de coördinaten van de volgende letterplaats bepaald. Pas als dit bekend is wordt de ruimte achter de letter opgevuld tot de volgende letterplaats.



Onderdelen grafisch TV display

De *printen* voor het grafisch display waren in korte tijd geheel uitverkocht. Inmiddels is echter weer een nieuwe

partij aangemaakt. Het is mogelijk dat u uw geld terug gestort kreeg met de mededeling 'uitverkocht'. In dat geval kunt u de printen gewoon opnieuw bestellen. De *dynamische RAM's* gaven ook wat problemen, de voorraden van zowel Intel als NEC bleken niet voldoende om de behoefte te dekken. Wij hebben Sprint Elektronika bereid gevonden de μ D411D in voorraad te nemen.

afb. 5 Met dit deel van het stroomdiagram wordt de ruimte onder en boven de letters in de achtergrond getekend.

afb. 6 Het laatste deel. Hier wordt de volgende letterpositie berekend. Het programma. De vet gedrukte adressen zijn aangepast op de I/O van de KIM, en moeten voor andere systemen worden aangepast. Wanneer een ander startadres gewenst wordt moeten de instructies op 07D4, 07E5, 07FE en 0838 aangepast worden.





LIJST 1

0600 D8	START	CLD				
0601 A5 A5		LDA KLEUR				
0603 29 18		AND #S18				
0605 85 BA		STA VOOR				
0607 A5 A5		LDA KLEUR				
0609 29 04		AND #S04				
060B 0A		ASL			Initialiseren	
060C 09 10		ORA #S10				
060E 85 BB		STA ACHTER				
0610 A2 00		LDX #S00				
0612 86 B2		STX VLAG				
0614 A5 A5		LDA KLEUR			Gegevens bekend?	
0616 29 01		AND #S01			Zo ja, naar 'BEKEND'.	
0618 D0 2C		BNE BEKEND				
061A A5 D2		LDA KARLO				
061C 85 C0		STA SUMLO			Kopieer het	
061E A5 D3		LDA KARHI			beginadres.	
0620 85 C1		STA SUMHI				
0622 A4 A0		LDY ASCII				
0624 B1 D4		LDA (TBLO), Y			Stel 'MODE' in.	
0626 29 E0		AND #SE0				
0628 85 C2		STA MODE				
062A B1 D4		LDA (TBLO), Y			Stel 'LENGTE' in.	
062C 29 F1		AND #S1F				
062E 85 C3		STA LENGTE				
0630 88	TELOP	DEY			Eén karakter terug.	
0631 C0 1F		CPY #S1F			Indien laatste	
0633 F0 11		BEQ BEKEND			karakter naar 'BEKEND'.	
0635 B1 D4		LDA (TBLO), Y			Karakterlengte	
0637 29 F1		AND #S1F			uit de tabel.	
0639 18		CLC				
063A 65 C0		ADC SUMLO				
063C 85 C0		STA SUMLO			Tel alle lengten bij elkaar	
063E A9 00		LDA #S00			op om het startadres van	
0640 65 C1		ADC SUMHI			het karakter te vinden.	
0642 85 C1		STA SUMHI				
0644 90 EA		BCC TELOP			Onvoorwaardelijk terug.	
0646 A0 00	BEKEND	LDY #S00				
0648 A5 A2		LDA YCOR				
064A 85 BC		STA YGRL				
064C A6 A4		LDX HOOG				
064E 38		SEC				
064F E9 08	LOOP	SBC #S08			Bepaald YGRL en YGRHI.	
0651 E6 BC		INC YGRL				
0653 E6 BC		INC YGRL				
0655 CA	DEX					
0656 D0 F6		BNE LOOP				
0658 85 BD		STA YGRH				
065A A9 20		LDA #S20				
065C 24 C2		BIT MODE				
065E F0 06		BEQ VERDER			Indien geschoven	
0660 30 04		BMI VERDER			YCOR goed zetten.	
0662 A5 BC		LDA YGRL				
0664 85 A2		STA YCOR				
0666 AD 02 17	WACHT1	LDA STUUR			Wacht op syncpuls	
0669 10 FB		BPL WACHT1			van de TV.	
066B A5 A1		LDA XCOR				
066D 8D 00 17		STA ADRES				
0670 EE 02 17		INC STUUR			X en Y-adres in het	
0673 EE 02 17		INC STUUR			display zetten.	
0676 A5 A2		LDA YCOR				
0678 8D 00 17		STA ADRES				
067B 8E 02 17		STX STUUR				
067E A5 A1		LDA XCOR				
0680 85 B8		STA XCOR2				
0682 A5 A2		LDA YCOR			Hulpcoördinaten	
0684 85 B6		STA YCOR2			instellen voor	
0686 85 B7		STA YCOR3			rechtopstaand figuur	
0688 A5 A3		LDA BREED				
068A 85 B4		STA BREED2				
068C A5 A4		LDA HOOG				
068E 85 B3		STA HOOG2				
0690 50 16		BVC ANDERE			Spring indien niet	
					gedraaid	
0692 A5 A1		LDA XCOR				
0694 49 FF		EOR #SFF				
0696 85 B6		STA YCOR2				
0698 85 B7		STA YCOR3				
069A A5 A2		LDA YCOR			Hulpcoördinaten	
069C 49 FF		EOR #SFF			instellen voor	
069E 85 B8		STA XCOR2			gekanteld figuur	
06A0 A5 A3		LDA BREED				
06A2 85 B3		STA HOOG2				
06A4 A5 A4		LDA HOOG				
06A6 85 B4		STA BREED2				
06A8 A5 B4	ANDERE	LDA BREED2			Stel de breedte	
06AA 85 B5		STA BREED3				
06AC A9 08	ZELFDE	LDA #S08				
06AE 85 B9		STA TEL				
06B0 B1 C0		LDA (SUMLO), Y			Zet het te tekenen	
06B2 85 B1		STA TEMP			byte in 'TEMP'.	
06B4 A6 B3	ANDBIT	LDX HOOG2			Stel hoogte in.	
06B6 24 B4		BIT VLAG				
06B8 70 04		BVS RECHTS			Schuif een bit	
06BA 06 B1		ASL TEMP			uit 'TEMP'	
06BC 50 02		BVC NORM			Normaal naar links	
06BE 46 B1	RECHTS	LSR TEMP				
06C0 24 A5	NORM	BIT KLEUR			Voor achtergrondkleur	
06C2 90 5A		BCC ACHTGR			naar 'ACHTGR'.	
06C4 10 61		BPL NIET			Voorg. tekenen?	
06C6 A5 BA		LDA VOOR			Stel voorgrond in	
06C8 8D 02 17		STA STUUR				
06CB AD 02 17	WACHT2	LDA STUUR			Wacht op syncpuls	
06CE 10 FB		BPL WACHT2			van de TV.	
06D0 24 C2		BIT MODE			Normaal naar 'YKLOK'	
06D2 50 5A		BVC YKLOK			om Y in te klokken.	
06D4 A9 FF		LDA #SFF				
06D6 45 B6		EOR YCOR2				
06D8 8D 00 17		STA ADRES			Klok het X-adres in.	
06DB EE 02 17		INC STUUR				
06DE A9 02		LDA #S02				
06E0 0D 02 17		ORA STUUR				
06E3 8D 02 17		STA STUUR			Schrijfpuls	
06E6 29 18	RUST	AND #S8			Terug naar de	
06E8 8D 02 17		STA STUUR			rusttoestand	
06EB C6 B6		DEC YCOR2			Als hetzelfde punt	
06ED CA		DEX			nog een keer moet,	
06EE D0 DB		BNE WACHT2			naar 'WACHT2'.	
06F0 C6 B9	NIBIT	DEC TEL			Zijn we al bij het	
06F2 D0 C0		BNE ANDBIT			laatste bitje?	
06F4 A5 B7		LDA YCOR3			Zet de coördinaten goed	
06F6 85 B6		STA YCOR2			voor de volgende kolom.	
06F8 E6 B8		INC XCOR2				
06FA AD 02 17	WACHT3	LDA STUUR			Wacht op syncpuls	
06FD 10 FB		BPL WACHT3			van de TV.	
06FF A5 B8		LDA XCOR2				
0701 8D 00 17		STA ADRES			Klok het nieuwe	
0704 EE 02 17		INC STUUR			X-adres in.	
0707 EE 02 17		INC STUUR				
070A 49 FF		EOR #SFF			Klok het gedraaide	
070C 8D 00 17		STA ADRES			X-adres in.	
070F A9 00		LDA #S00				
0711 8D 02 17		STA STUUR				
0714 C6 B5		DEC BREED3			Als hetzelfde byte nog	
0716 D0 94		BNE ZELFDE			n keer moet: naar 'ZELFDE'.	
0718 F0 26		BEQ NIBYTE			Onvoorwaardelijk	
071A F0 8C	ANDER2	BEQ ANDERE			Tussensprong voor	
071C D0 8A		BNE ANDERE			long branches	
071E 50 07	ACHTGR	BVC NIET			Achtergr. tekenen?	
0720 A5 B8		LDA ACHTER				
0722 8D 02 17		STA STUUR			Stel achtergrond in.	
0725 70 A4		BVS WACHT2			Onvoorwaardelijk	
0727 C6 B6	NIET	DEC YCOR2				
0729 CA		DEX			Verplaats de pointer	
072A D0 FB		BNE NIET			zonder te tekenen.	
072C F0 C2		BEQ NIBIT			Onvoorwaardelijk	
072E A5 B6	YKLOK	LDA YCOR2				
0730 8D 00 17		STA ADRES			Klok het Y-adres	
0733 A9 02		LDA #S02			in.	
0735 0D 02 17		ORA STUUR				
0738 8D 02 17		STA STUUR				
073B EE 02 17		INC STUUR			Schrijfpuls	
073E D0 A6		BNE RUST			Onvoorwaardelijk	
0740 24 B2	NIBYTE	BIT VLAG			Bij spiegelen verder	
0742 30 1C		BMI SYMR			op 'SYMR'.	
0744 C8		INY			Volgende byte	
0745 C4 C3		CPY LENGTE			Zolang laatste byte	
0747 D0 D1		BNE ANDER2			niet is geweest: terug.	
0749 24 B2		BIT VLAG			Als we klaar zijn met	
074B 70 3F		BVS KLAAR			spiegelen, of als er	
074D 24 C2		BIT MODE			geen symmetrie gevraagd	
074F 10 3B		BPL KLAAR			wordt naar 'KLAAR'.	
0751 88		DEY				
0752 A9 80		LDA #S80			Vlag op rechte	
0754 85 B2		STA VLAG			symmetrie.	
0756 A9 20		LDA #S20				
0758 24 C2		BIT MODE			Als het figuur is	
075A 70 09		BVS KANT			gekanteld, naar 'KANT'.	
075C D0 02		BNE SYMR				
075E F0 BA		BEQ ANDER2			Herhaal het proces	
0760 88	SYMR	DEY			Vorige byte	
0761 10 B7		BPL ANDER2			Terug indien niet klaar.	



SOFTWARE grafisch display

0763 30 27		BMI KLAAR	Onvoorwaardelijk
0765 A5 B7	KANT	LDA YCOR3	
0767 A6 A3		LDX BREED2	
0769 38		SEC	
076A E9 08	COR	SBC # \$08	Coördinaten instellen op volgende aansluitende karakterplaats.
076C CA		DEX	
076D D0 FB		BNE COR	
076F 85 B6		STA YCOR2	
0771 85 B7		STA YCOR3	
0773 A5 A2		LDA YCOR	
0775 8D 00 17		STA ADRES	
0778 A2 02		LDX # \$02	
077A 8E 02 17		STX STUUR	Nieuwe adressen inklokken.
077D A0 00		LDY # \$00	
077F 8C 02 17		STY STUUR	
0782 49 FF		EOR # \$FF	
0784 85 B8		STA XCOR2	XCOR2 goed zetten
0786 A9 40		LDA # \$40	
0788 85 B2		STA VLAG	Zet 'VLAG' op gekantelde symmetrie
078A D0 9E		BNE ANDER2	
078C A5 C3	KLAAR	LDA LENGTE	Lengte in 'TEMP'
078E 85 B1		STA TEMP	
0790 24 C2		BIT MODE	
0792 50 06		BVC TEST	
0794 08		PHP	Indien gekanteld: karakterlengte is 8
0795 A9 08		LDA # \$08	
0797 85 B1		STA TEMP	
0799 28		PLP	
079A 10 0A	TEST	BPL GEWOON	Is er symmetrie?
079C 06 B1		ASL TEMP	
079E A9 20		LDA # \$20	Bij symm. lengte x 2
07A0 24 C2		BIT MODE	
07A2 F0 02		BEQ GEWOON	Bij enkele middenpoot lengte - 2
07A4 C6 B1		DEC TEMP	
07A6 A6 A3	GEWOON	LDX BREED	
07A8 A5 A1		LDA XCOR	
07AA 18	LOOP1	CLC	
07AB 65 B1		ADC TEMP	
07AD CA		DEX	Bereken de nieuwe X-coördinaat, en zet hem in 'TEMP'
07AE D0 FA		BNE LOOP1	
07B0 85 B1		STA TEMP	
07B2 24 A5		BIT KLEUR	
07B4 50 4B		BVC GEACHT	
07B6 A5 A1		LDA XCOR	
07B8 85 B8		STA XCOR2	Coördinaten van het te vullen vlak.
07BA A5 BD		LDA YGRH	
07BC 85 B7		STA YCOR3	
07BE 24 C2		BIT MODE	
07C0 50 28		BCV NIEKAN	Indien niet gekanteld naar 'NIEKAN'
07C2 A5 C3		LDA LENGTE	
07C4 C9 08		CMP # \$08	Indien hoger dan 8 maar 'NIEVUL'
07C6 B0 0F		BCS NIEVUL	
07C8 A5 A2		LDA YCOR	
07CA A6 A4		LDX HOOG	
07CC 38	LOOP2	SEC	
07CD E5 C3		SBC LENGTE	
07CF CA		DEX	Bereken de gekantelde hoogte
07D0 D0 FA		BNE LOOP2	
07D2 85 B6		STA YCOR2	
07D4 20 58 08		JSR TKNBLK	
07D7 A5 BC	NIEVUL	LDA YGRL	Teken boven het kar. Indien niet geschoven
07D9 C5 A2		CMP YCOR	
07DB F0 24		BEQ GEACHT	strook onder het karakter tekenen.
07DD A5 BC	NIGESC	LDA YGRL	
07DF 85 B6		STA YCOR2	
07E1 A5 A2		LDA YCOR	
07E3 85 B7		STA YCOR3	Vul de strook onder het karakter.
07E5 20 58 08		JSR TKNBLK	
07E8 F0 17		BEQ GEACHT	Onvoorwaardelijk
07EA A5 BC	NIEKAN	LDA YGRL	
07EC C5 A2		CMP YCOR	Indien niet geschoven onderstrook gaan tekenen.
07EE D0 ED		BNE NIGESC	
07F0 A5 A4		LDA HOOG	
07F2 0A		ASL	
07F3 0A		ASL	Bereken karakterhoogte.
07F4 0A		ASL	
07F5 85 B3		STA HOOG2	
07F7 38		SEC	
07F8 A5 A2		LDA YCOR	Vul de strook boven het karakter.
07FA E5 B3		SBC HOOG2	
07FC 85 B6		STA YCOR2	
07FE 20 58 08		JSR TKNBLK	
0801 A5 BC	GEACHT	LDA YGRL	Was het karakter geschoven?
0803 C5 A2		CMP YCOR	
0805 D0 09		BNE VERDER	
0807 A6 A4		LDX HOOG	
0809 C6 A2	LOOP3	DEC YCOR	

080B C6 A2		DEC YCOR	Zet YCOR weer op de oude waarde.
080D CA		DEX	
080E D0 F9		BNE LOOP3	
0810 A5 B1	VERDER	LDA TEMP	
0812 85 B8		STA XCOR2	Bereken nieuwe letterplaats voor variabele letterafstand
0814 18		CLC	
0815 65 A6		ADC SPATIE	
0817 85 B1		STA TEMP	
0819 A9 02		LDA # \$2	Moet er vaste letterafstand ingesteld worden?
081B 24 A5		BIT KLEUR	
081D F0 09		BEQ VARI	
081F 08		PHP	
0820 A5 A1		LDA XCOR	Bereken de nieuwe letterplaats voor vaste letterafstand
0822 18		CLC	
0823 65 A7		ADC LTRAF	
0825 85 B1		STA TEMP	
0827 28		PLP	
0828 A5 B8	VARI	LDA XCOR2	
082A C5 B1		CMP TEMP	Ruimte tussen de letters opvullen?
082C B0 0D		BCS OVER	
082E 50 08		BVC OVER	
0830 A5 BC		LDA YGRL	
0832 85 B6		STA YCOR2	Vul het vlak vanaf laatste karakter tot de nieuwe letterplaats.
0834 A5 BD		LDA YGRH	
0836 85 B7		STA YCOR3	
0838 20 58 08		JSR TKNBLK	
083B A5 B1	OVER	LDA TEMP	Is de achter-kantlijn gepasseerd?
083D C5 D0		CMP KNTLAC	
083F 90 14		BCC FINE	
0841 A9 20		LDA # \$20	
0843 24 A5		BIT KLEUR	Automatisch CR en LF?
0845 F0 10		BEQ KLAAR	
0847 A6 A4		LDX HOOG	
0849 A5 A2		LDA YCOR	
084B 18	LOOP4	CLC	Verzorg een LF.
084C 69 0A		ADC # \$0A	
084E CA		DEX	
084F D0 FA		BNE LOOP4	
0851 85 A2		STA YCOR	Verzorg een CR
0853 A5 D1		LDA KNTLVO	
0855 85 A1	FINE	STA XCOR	Terug naar hoofdprog.
0857 60	KLAAR	RTS	
0858 A5 BB	TKNBLK	LDA ACHTER	Stel begincoördinaten in.
085A 8D 02 17		STA STUUR	
085D A4 B6		LDY YCOR2	
085F A6 B8		LDX XCOR2	
0861 AD 02 17	LOOP5	LDA STUUR	
0864 10 FB		BPL LOOP5	
0866 8E 00 17		STX ADRES	Vul een vlak, begrensd door XCOR2, TEMP YCOR2 en YCOR3.
0869 EE 02 17		INC STUUR	
086C EE 02 17		INC STUUR	
086F 8C 00 17		STY ADRES	
0872 EE 02 17		INC STUUR	
0875 8D 02 17		STA STUUR	
0878 E8		INX	
0879 E4 B1		CPX TEMP	
087B D0 E4		BNE LOOP5	
087D 88		DEY	
087E C4 B7		CPY YCOR3	
0880 D0 DD		BNE LOOP5	
0882 60		RTS	Terug.

Letters op het grafisch TV-display

Unieke softwareservice

RB komt (voorlopig experimenteel) met een unieke softwareservice. In het radioprogramma Hobbyscoop van woensdag 6 juni zal het hier besproken programma worden uitgezonden.

Wij zullen hierbij het KIM-formaat hanteren, waardoor alle KIM, AIM en SYM (of VIM) bezitters gratis in het bezit kunnen komen van dit programma (Intikken kan natuurlijk ook).

Het best kunt u de cassetterecorder welke u bij de computer gebruikt op de radio aansluiten en het programma opnemen. Hobbyscoop wordt elke woensdagavond uitgezonden om 22.30 uur op Hilversum 1.